







ROBÓTICA

Módulo 2



Robô Ultrassônico

AULA

10

KIT 2023

GOVERNADOR DO ESTADO DO PARANÁ

Carlos Massa Ratinho Júnior

SECRETÁRIO DE ESTADO DA EDUCAÇÃO

Roni Miranda Vieira

DIRETOR DE TECNOLOGIA E INOVAÇÃO

Claudio Aparecido de Oliveira

COORDENADOR DE TECNOLOGIAS EDUCACIONAIS

Marcelo Gasparin

Produção de Conteúdo

Cleiton Rosa

Adilson Carlos Batista

Validação de Conteúdo

Cleiton Rosa

Revisão Textual

Kellen Pricila dos Santos Cochinski

Projeto Gráfico e Diagramação

Edna do Rocio Becker

2024

Aula 02	Arduíno: Bibliotecas e Funções	Aula 01	O que já vimos?
Aula 04	Semáforo Inteligente com IR	Aula 03	Código Morse
Aula 06	Matriz de LED 8X8	Aula 05	Semáforo Completo com Display
Aula 08	Painel de Senhas	Aula 07	Desenhando na matriz de LEDs
Aula 10	Robô Autônomo	Aula 09	Escrevendo mensagens
Aula 12	Sensor de Umidade do Solo	Aula 11	Sensor de Chuva
Aula 14	Feedbacks + Inventário I	Aula 13	Irigador Automático
Aula 16	Servos Motores	Aula 15	Teclado Matricial de Membrana
Aula 18	Controlando Servos Motores	Aula 17	Fechadura Eletrônica
Aula 20	Braço Robótico	Aula 19	JoyStick Shield
Aula 22	Sensor de Som	Aula 21	Sensor de Movimento Presença
Aula 24	Termômetro Digital	Aula 23	Sensor de Umidade e Temperatura
Aula 26	Acelerômetro e Giroscópio	Aula 25	Sensor de Gás e Fumaça
Aula 28	Feedbacks + Inventário II	Aula 27	Motor de Passo
Aula 30	Relé	Aula 29	Receptor IR e Controle Remoto
Aula 32	Módulo RF 433mhz - II	Aula 31	Módulo RF 433mhz - I
Aula 34	Módulo Wireless	Aula 33	Projeto CHAT via RF
Aula 36	Módulo WI-FI - IoT com Sensores	Aula 35	Comunicação do Módulo WI-FI em HTML
Aula 38	Módulo WI-FI - IoT com Atuadores (Relé)	Aula 37	Módulo WI-FI - IoT com Atuadores (LED)
Aula 40	Monitor de Sensores em HTML II	Aula 39	Monitor de Sensores em HTML I
Aula 42	Feedbacks + Inventário III	Aula 41	Mostra de Robótica

Aula 09
Escrevendo Mensagens

Aula 10 Robô Ultrassônico

Aula 11
Sensor de Chuva

Sumário

Introdução	2
Objetivos desta aula	3
Lista de materiais	4
Roteiro da aula	5
1. Contextualização	5
2. Montagem e programação	9
3. Feedback e finalização	20



INTRODUÇÃO

Atualmente, uma variedade de modelos de robôs é empregada para tarefas simples, como a limpeza de residências, corredores, espaços amplos em empresas e indústrias, entre outras atividades. Esses robôs frequentemente se deparam com diversos obstáculos durante suas operações, porém, têm a capacidade de desviar deles de maneira natural. Já se perguntou como eles realizam essa proeza?

Nesta aula, vamos explorar justamente isso, reutilizando o **Sensor de Distância Ultrassônico HC-SR04** para criar um protótipo. Nosso objetivo é programar um robô capaz de desviar de obstáculos de forma autônoma, mergulhando no fascinante mundo da robótica.







OBJETIVOS DESTA AULA

- Rever e compreender o funcionamento do sensor de distância ultrassônico HC-SR04;
- Criar um protótipo funcional de robô ultrassônico;
- Programar o robô utilizando o sensor de distância ultrassônico HC-SR04 para que ele possa identificar obstáculos e evitar colisões de forma autônoma;
- Incentivar a exploração e a concepção de novos projetos que envolvam a aplicação de robôs autônomos, sensores e suas diversas funcionalidades, promovendo assim o desenvolvimento criativo e prático no campo da robótica.

LISTA DE MATERIAIS

- 01 placa Arduino Uno R3;
- 01 cabo USB;
- 01 sensor de distância ultrassônico HC-SR04;
- 01 suporte para sensor ultrassônico;
- 01 ponte HL298n (driver);
- 01 kit chassi 2WD robô para Arduino;
- 02 jumpers macho-macho;
- 08 jumpers macho-fêmea;
- 01 fonte de energia (pilhas, bateria 9V ou fonte chaveada);
- 01 chave de fenda pequena;
- Parafusos ou fita adesiva para fixação de componentes ao chassi;
- Notebook;
- Software Arduino IDE.



ROTEIRO DA AULA

1. Contextualização

Hoje em dia, uma variedade de modelos de robôs é amplamente empregada para realizar tarefas simples, tanto em ambientes industriais e empresariais quanto em residências. Provavelmente, você já ouviu falar ou até mesmo viu algum deles em ação! Por exemplo, há os aspiradores de pó autônomos, frequentemente anunciados em comerciais de lojas. Além de aspirar a sujeira, esses dispositivos são capazes de desviar de obstáculos, como ilustrado na figura 1. Outro exemplo são os robôs cortadores de grama, representados na figura 2, que operam de maneira similar, navegando sistematicamente pelo ambiente.

Nas empresas, especialmente as de grande porte, e nas indústrias, é comum encontrar robôs encarregados da limpeza de salas e corredores, todos seguindo o mesmo princípio operacional. Você pode estar se perguntando: o que todos esses robôs têm em comum? A resposta é simples: eles fazem uso de sensores ultrassônicos.

Figura 1 - Robô aspirador



Figura 2 - Cortador de grama aspirador



Esses sensores emitem ondas sonoras de alta frequência, as quais são refletidas ao encontrar um obstáculo. Essa informação permite que o robô calcule a distância até o obstáculo e, assim, desvie dele para realizar sua tarefa com precisão. Além disso, o sensor ultrassônico auxilia o robô a corrigir sua trajetória quando necessário.

Em nosso projeto, trabalharemos com o **Sensor de Distância Ultrassônico HC-SR04**, como ilustrado na figura 3. Você já teve contato com esse modelo de sensor em diversas aulas anteriores, incluindo:

- Aula 35 – Sensor de Estacionamento;
- Aula 38 – Robô Sumô – Estrutura;
- Aula 39 – Robô Sumô – Programação + Treinamento I;
- Aula 40 – Robô Sumô – Programação + Treinamento II, todas do Módulo I.





Figura 3 - Sensor de Distância Ultrassônico HC-SR04

**ESPECIFICAÇÕES:****Alimentação:** 5V DC**Corrente de Operação:** 2mA**Ângulo de efeito:** 15°**Alcance:** 2cm a 4m**Precisão:** 3mm

Relembrando como o Sensor de Distância Ultrassônico funciona.


O sensor emite ondas sonoras de frequência ultrassônica, as quais, ao encontrarem objetos, são refletidas em direção ao sensor. A **distância entre o sensor e o objeto que reflete a onda é calculada com base no tempo decorrido entre o envio e a leitura do retorno**, levando em conta a velocidade do som, que é praticamente constante no ar (cerca de 340 m/s). Vale ressaltar que essas ondas ultrassônicas não são audíveis pelo ouvido humano.

A ativação do pulso ultrassônico ocorre através do **pino Trig**, enquanto a leitura do retorno é realizada pelo **pino Echo**. Além desse sensor, outro componente fundamental é o **Kit Chassi 2WD Robô para Arduino**, conforme mostrado na figura 4. Você já teve a oportunidade de trabalhar com esse kit em aulas anteriores sobre Robô Sumô, juntamente com outros componentes essenciais, como o Arduino, que desempenha um papel indispensável em muitos dos projetos que estudamos até o momento.

Figura 4 - Kit Chassi 2WD Robô para Arduino

Especificações:

Chassi em acrílico
Motores DC (3 a 6v)
Rodas de Borracha
Roda Boba (Universal)
Discos de Encoder
Suporte para 4 Pilhas
Jogo de Parafusos



Como você já teve a oportunidade de estudar em aulas anteriores, o Kit Chassi é a base fundamental para a construção do nosso robô. Ele apresenta uma estrutura de acrílico com furos estrategicamente posicionados para a fixação de equipamentos e peças utilizando parafusos. Além disso, o chassi é equipado com três rodas: duas de borracha para tração e uma rod wheel universal para possibilitar os giros necessários. Também inclui um suporte para acomodar as pilhas e dois motores DC responsáveis por acionar as rodas.

Agora que revisitamos os componentes essenciais para o nosso robô ultrassônico, estamos prontos para avançar para a emocionante etapa de montagem e programação!

2. Montagem e programação

Antes de começarmos a montagem do nosso robô autônomo, é crucial garantir que o kit chassi 2WD

esteja montado corretamente. Isso significa que os motores e as rodas devem estar devidamente fixados à estrutura de acrílico. Se você precisar montar ou fazer ajustes, pode consultar o guia de montagem do chassi 2WD através do link abaixo:

[Manual de montagem do kit chassi 2WD](#)

Para iniciar a montagem do seu robô autônomo, comece fixando os componentes sobre o chassi utilizando parafusos e/ou fita adesiva, de acordo com as posições indicadas na Figura 4.

É importante tomar cuidado para evitar que as placas entrem em contato com os parafusos metálicos do chassi, pois isso pode danificar os circuitos. Uma sugestão é usar fita adesiva sobre os parafusos para isolá-los dos componentes eletrônicos.


Com o auxílio de uma pequena chave fenda, fixe os dois pares de fios dos motores nos dois bornes da ponte H. Não se preocupe com a ordem dos fios, pois mais adiante,



quando iniciarmos a programação, faremos os testes para averiguar se algum dos motores está girando no sentido contrário dos comandos, sendo assim, basta alterar a posição desses fios.

Certifique-se de que todas as placas estejam devidamente isoladas para evitar interferências indesejadas durante o funcionamento do robô. Seguindo essas instruções, você estará pronto para continuar com a montagem do seu robô autônomo de maneira segura e eficiente.

Figura 5: Encaixe das placas sobre o Chassi



Na sequência, utilizando uma chave de fenda, fixe dois jumpers macho-macho nos bornes de alimentação da ponte H (**12V** e **GND**). Em seguida, conecte as outras extremidades dos jumpers nas portas de alimentação do Arduino (**Vin** e **GND**, respectivamente), conforme ilustrado na figura 5.

Após isso, utilizando 4 jumpers macho-fêmea, conecte os pinos IN1 ao IN4 da ponte H com as portas digitais PWM do Arduino, conforme descrito na tabela 1.

Tabela 1 - Conexão entre a ponte H e o Arduino

Pinos Ponte H	IN1	IN2	IN3	IN4
Portas PWM Arduino	~10	~9	~6	~5

Atenção especial deve ser dada aos pinos **IN1** a **IN4** da ponte H, pois é crucial conectá-los às portas digitais que suportam PWM (identificadas pelo símbolo ~). Essa configuração possibilitará o controle eficaz da potência dos dois motores do robô.






Figura 6: Conexão da ponte H ao Arduino



Agora vamos realizar a conexão do Sensor Ultrassônico ao Arduino. Para isso, você precisará de 4 jumpers macho-fêmea. Conecte os pinos **GND**, **VCC**, **Echo** e **Trig** do sensor nas portas de alimentação **GND** e **5V**, e nas portas digitais **11** e **12** do Arduino, respectivamente. Essa disposição está ilustrada na figura 7.

Figura 7: Conexão do Sensor Ultrassônico ao Arduino




Por fim, para concluir a montagem, conecte o cabo USB e a fonte de energia (pilhas ou bateria) ao conector jack P4 da placa Arduino. É fundamental estar atento à fonte de energia utilizada, uma vez que os motores demandam um consumo significativo durante o funcionamento. Portanto, é essencial empregar uma fonte capaz de fornecer uma tensão entre 6 e 12 Volts.

Um exemplo prático seria o uso de um estojo com quatro pilhas 18650, conforme mostrado na figura 8. No entanto, você tem a flexibilidade de es-

colher qualquer outra fonte de energia que atenda aos requisitos mínimos de operação do seu projeto.

Sugestão: Entre o plugue P4 e a bateria, você pode adicionar uma pequena chave interruptora. Isso facilitará o ligamento e desligamento do seu robô, proporcionando uma maneira conveniente de controlar a alimentação. Essa chave interruptora pode ser facilmente adicionada ao circuito, permitindo uma operação mais intuitiva e segura do seu projeto.

Figura 8: Conexão da bateria e cabo USB ao robô autônomo



Com a montagem dos componentes eletrônicos concluída, o próximo passo é programar o Arduino para controlar os movimentos do robô autônomo. Este processo envolve a criação de um código que irá ler os dados do sensor ultrassônico e acionar os motores de acordo com as informações recebidas. Vamos agora mergulhar na fase de programação para tornar o nosso robô autônomo totalmente funcional e pronto para operação.

Agora, vamos programar!

Com os componentes eletrônicos montados, é hora de programar nosso robô por meio de codificação. Vamos criar o código que irá controlar o funcionamento do robô, desde a leitura dos dados do sensor ultrassônico até o acionamento dos motores para desviar de obstáculos e realizar as ações desejadas. Este é um passo crucial para garantir que nosso robô autônomo opere de forma eficaz e inteligente. Vamos começar a escrever o código para dar vida ao nosso projeto!



Linguagem de programação por codificação

Para iniciar a programação, é necessário ter a biblioteca **“Ultrasonic.h”** instalada no Software Arduino IDE. Você pode encontrar orientações sobre como instalar bibliotecas na **Aula 05 - Softwares Arduino IDE e mBlock**.

Uma vez que as bibliotecas estejam instaladas, conecte a placa Arduino Uno ao computador utilizando um cabo USB para estabelecer a comunicação entre a placa microcontroladora e o software Arduino IDE.

No software IDE, escreva o código-fonte de programação ou copie e cole conforme apresentado no quadro 1:

Quadro 1 - Código-fonte da programação na linguagem do Arduino (Wiring)

```

/*****
/* Aula 10 - Robô Ultrassônico */
/* Programação do projeto robô ultrassônico. */
/* Ao transferir o código abaixo para seu Arduino, o robô */
/* será capaz de mover para frente e ao se aproximar de */
/* obstáculo, irá realizar uma manobra para desviá-lo. */
/* Links para obtenção da biblioteca by Erick Simões */
/* http://librarymanager/All#minimalist#Ultrasonic */
/* https://github.com/ErickSimoes/Ultrasonic */
*****/

#include <Ultrasonic.h>
/* Define os pinos para o sensor ultrassônico. */
#define pino_Trig 11
#define pino_Echo 12
/* Define a potência dos motores: 0 a 255. */
int potencia = 200;

/* Definições dos pinos IN1 a IN4 da ponte H. */
/* Para controlar a velocidade dos motores, esses pinos */
/* deverão ser conectados em portas com o recurso PWM. */
#define IN1 10 // Porta PWM ~10
#define IN2 9 // Porta PWM ~9
#define IN3 6 // Porta PWM ~6
#define IN4 5 // Porta PWM ~5

/* Cria o objeto Sensor que realizará a leitura do sensor. */
Ultrasonic Sensor(pino_Trig, pino_Echo);
/* Variável que armazenará as medidas. */
int distancia;

/* Criando as funções de controle dos motores. */

/* Move para frente. */
void Frente(int Vel_A, int Vel_B) {
  /* Motor A */
  analogWrite(IN1, Vel_A);
  analogWrite(IN2, 0);
  /* Motor B */
  analogWrite(IN3, Vel_B);
  analogWrite(IN4, 0);
}

/* Move vpara trás. */
void Re(int Vel_A, int Vel_B) {
  /* Motor A */
  analogWrite(IN1, 0);
  analogWrite(IN2, Vel_A);
}

```

```
/* Motor B */
analogWrite(IN3, 0);
analogWrite(IN4, Vel_B);
}

/* Giro no sentido horário. */
void GiroHorario(int vel_A, int vel_B) {
  /* Motor A */
  analogWrite(IN1, vel_A);
  analogWrite(IN2, 0);
  /* Motor B */
  analogWrite(IN3, 0);
  analogWrite(IN4, vel_B);
}

/* Giro no sentido anti-horário. */
void GiroAntiHorario(int Vel_A, int Vel_B) {
  /* Motor A */
  analogWrite(IN1, 0);
  analogWrite(IN2, Vel_A);
  /* Motor B */
  analogWrite(IN3, Vel_B);
  analogWrite(IN4, 0);
}

/* Para o chassi. */
void Pare() {
  /* Motor A */
  analogWrite(IN1, 0);
  analogWrite(IN2, 0);
  /* Motor B */
  analogWrite(IN3, 0);
  analogWrite(IN4, 0);
}

/* Função parar manobrar o robô. */
void Manobre() {
  if (random(1, 3) == 1) {
    GiroAntiHorario(potencia * 0.5, potencia * 0.5);
  } else {
    GiroHorario(potencia * 0.5, potencia * 0.5);
  }
  delay(250);
}

void setup() {
  /* Inicia com os motores parados. */
  Pare();
  /* Aguarda 3 segundos antes de começar os movimentos. */
```

```
    delay(3000);
}

void loop() {
    /* Armazena a distância na variável "distancia". */
    distancia = Sensor.read();
    /* Se a distância for menor ou igual a 40cm, realize a */
    /* manobra do robô. */
    if (distancia <= 40) {
        /* Chama a função responsável pela manobra. */
        Manobre();
    }
    /* Senão siga em frente. */
    else {
        /* Chama a função mover para frente. */
        Frente(potencia, potencia);
    }
}
```

Após inserir o código-fonte no Arduino IDE, compile e/ou carregue o programa para o Arduino.

Após a transferência do programa para o Arduino, o robô aguardará 3 segundos para iniciar o seu movimento. O sensor ultrassônico estará constantemente determinando a distância de objetos ou paredes à sua frente enquanto o robô se move para frente. Quando a distância for menor ou igual a 40 centímetros, o robô interromperá o movimento, realizará a manobra e, em seguida, retomará seu movimento para frente. Esta sequência será repetida infinitamente até que o robô seja desligado.

Desafio:

E que tal ir além? Que tal acoplar o sensor ultrassônico a um suporte fixado a um micro servo? Dessa forma, quando o robô se deparar com um obstáculo, ele poderá “olhar” para os dois lados e tomar a decisão de seguir pela direção menos obstruída. Isso proporcionaria ao robô uma capacidade ainda maior de navegação autônoma e adaptativa. Que tal aceitar esse desafio e levar seu robô autônomo a um novo patamar de funcionalidade e inteligência?

E se...?

Se você enfrentar problemas com o funcionamento do projeto, preste atenção a alguns dos possíveis erros:

- **Confira a conexão da ponte H com a placa Arduino:** verifique se a ponte H foi conectada corretamente aos pinos da placa Arduino.
- **Polaridade dos fios do motor:** verifique se os fios do motor não estão conectados de forma invertida, o que pode causar a rotação contrária do motor. Se necessário, inverta a ordem dos fios nos bornes da ponte H.
- **Confira a conexão do sensor:** verifique se o sensor está corretamente conectado às mesmas portas do Arduino que foram utilizadas na programação.
- **Verifique a fonte de alimentação:** certifique-se de que a fonte de alimentação externa utilizada tenha carga suficiente para manter o robô em funcionamento. Se necessário, faça a substituição da fonte de energia.

Ao identificar e corrigir qualquer um desses possíveis erros, seu projeto estará pronto para funcionar corretamente. Se ainda assim encontrar dificuldades, não hesite em buscar ajuda adicional ou revisar cuidadosamente as etapas do projeto.

3. Feedback e Finalização

- a. Confira, compartilhando seu projeto com os demais colegas, se o objetivo foi alcançado.
- b. Analise seu projeto desenvolvido, de modo a atender aos requisitos para funcionamento do robô autônomo.
- c. Reflita se as seguintes situações ocorreram:
 - i. Colaboração e cooperação: você e os membros de sua equipe interagiram entre si, compartilhando ideias que promoveram a aprendizagem e o desenvolvimento deste projeto?
 - ii. Pensamento crítico e resolução de problemas: você conseguiu identificar os problemas, analisar informações e tomar decisões de modo a contribuir para o projeto desenvolvido?
- d. Reúna todos os componentes utilizados nesta aula e os organize novamente, junto aos demais, no kit de robótica.



DIRETORIA DE TECNOLOGIAS E INOVAÇÃO (DTI)
COORDENAÇÃO DE TECNOLOGIAS EDUCACIONAIS (CTE)

EQUIPE ROBÓTICA PARANÁ

- Ailton Lopes
- Andrea da Silva Castagini Padilha
- Cleiton Rosa
- Darice Alessandra Deckmann Zanardini
- Edgar Cavalli Junior
- Edna do Rocio Becker
- José Feuser Meurer
- Kellen Pricila dos Santos Cochinski
- Marcelo Gasparin
- Michele Serpe Fernandes
- Michelle dos Santos
- Roberto Carlos Rodrigues

Os materiais, aulas e projetos da “Robótica Paraná”, foram produzidos pela Coordenação de Tecnologias Educacionais (CTE), da Diretoria de Tecnologia e Inovação (DTI), da Secretaria de Estado da Educação do Paraná (SEED), com o objetivo de subsidiar as práticas docentes com os estudantes por meio da Robótica. Este material foi produzido para uso didático-pedagógico exclusivo em sala de aula.



Este trabalho está licenciado com uma Licença
Creative Commons – CC BY-NC-SA
[Atribuição - NãoComercial - Compartilha Igual 4.0](https://creativecommons.org/licenses/by-nc-sa/4.0/)

